
Is Paradox a good front-end for MySql ?

Caution

Be aware that some functionalities of Paradox concerning non standard alias (ODBC and SQL) are broken with
the Service Pack 2 of windows XP. You can check if you are in XP SP2 by executing the command winver
(Menu Start/execute). If  the end of the version information is « Service Pack 2 » you are in SP2. The problem is
tied to some Corel dialog box with the alias listbox where you can not more select a non standard alias.

Examples of such dialog box: Menu File/Open /Table, Menu File /New/Query , Menu tools / Utilities…

Concerning SQL tables please note that the SQL editor is not concerned nor the Visual query builder. For a table
access we can build a form (not with the expert) or use Proview , for a QBE we can create a new datamodel and
then create a QBE based on that. OPAL access is not concerned by this problem.

Hopefully Corel will quickly provide a patch…
For this document I’ve worked in windows XP SP1

Installation of the mysql driver

- Find and install the package « MyOdbc » from mysql web site.

I ‘ve used MyOdbc standard 3.51.9 the latest stable version. There is a 3.52 for test. Due to some problem
I ‘ve found on Datetime field I’ve also tested the old 2.5 MyOdbc. You can work with both driver installed

- Now you can find the MySql Odbc driver in the windows Odbc panel
Menu Start / Parameter / Control Panel / Administration tools / Data Source (ODBC)

Create a DNS System entry

Verify in BDE Administrator this connection

Menu Start / Parameter / Control Panel / BDE Administrator.We can save here the DATABASE NAME,USER
NAME,SQLQRYMODE parameter.(You should also check Auto ODBC = True in configuration/system/init)

Firsts tests

Connecting from Paradox to an existing MySql table in an existing database

We can use PhpMyAdmin as our first MySql administration tool for a simple installation. Consider a simple
database named mydb with one table mytable and several fields :

Now open Paradox 10
Menu File / Open / Table , choose TestMySql in the alias box
You should now be able to open the mytable table…

Working with a MySql table

Hit <F9> and enter our first record :

Field Value Comment
ID 1 Is it possible to specify an

autoincrement value ? YES !
LastName Smith
First Name Paul
Birthday 25/06/1964 I enter the date in my French format

and… IT WORKS !
Amount 10000.53 Again I enter the number with  my

format and… IT WORKS !
Comment Trying to store a small text in a MySql

TEXT field. How this will work ?
Can we make a new paragraph ?

YES !

Is it possible to make a find/replace (right
mouse CLIC) ?
I ask to replace "clic" with "CLIC" and...
IT WORKS !

Now is it possible to see the blob content
each time I reach a record ?
Let see, Right mouse clic on the
Comment field / Property then check the
Blob content display and... IT WORKS !

<F2> <SHIFT F2> …
Easy isn’t it ?

LastUpdate (DateTime) 15:52:46, 14/10/2004 (space bar) An error occur : another user is
editing this record… Then I try to
delete the field and post the record
and a new error : field required …
Tested in MyOdbc 3.5 and 2.5

LastUpdate (Timestamp) 15:52:46, 14/10/2004
(by default and space bar )

The same error occur randomly
Tested in MyOdbc 3.5 and 2.5

LastUpdate (Time) 15:52:46 (space bar) Same error in 3.5, OK in 2.5

Any other problem ?

Well at least one, the Paradox project viewer can not deal with SQL alias (not related to the XP SP2 bug), well
we can use Proview a freeware/shareware from Vladimir Menkin http://members.fortunecity.com/menkin/

Classical operations on tables in interactive mode

Now let us compare Paradox and phpMyAdmin for common table tasks.
PhpMyAdmin is really good in that part…

Task Paradox PhpMyAdmin
Rename a table Not Possible for SQL table Possible
Copy a table An error occur (Incorrect field type)

(with MyODBC 2.52 the field type is not respected )
Possible

Empty a table Possible Possible
Delete a table Possible Possible
Sort by any column Not possible without a query Possible
Filter on alpha fields Possible Possible (Select) *
Filter on integer or numeric fields Possible Possible (Select) *
Filter on date fields Not possible without a query Possible (Select) *

* you don’t have to enter the SQL yourself

Quick transfert between Paradox and MySql tables

If your Field type are simple the easiest way is to use the Copy tool utility in Paradox :

Menu tool / Utilities / Copy

This tool let you choose a local alias for your Paradox tables and a SQL alias for your MySql ones.
Both way are handled : Paradox to MySql or MySql to Paradox.

Query with QBE

Paradox has here a little advantage, his QBE will generate an «ANSWER » table that can be reuse.
In phpMyAdmin by default you receive the result on screen. Not so simple to reuse the answer (but possible via
an export or a temp table). Another big advantage for Paradox , the ability to visualy join tables and to make
calculation or concatenation. But each new version of phpMyAdmin comes with real new features …

Task Paradox PhpMyAdmin
Selection of fields Possible Possible
Criteria > , < , = , >= , <= , LIKE Possible Possible
Criteria !=  (Not) Possible Possible
Sort on one field ASC , DESC Possible Possible
Sort on several fields Possible Possible
Change the fields order in the answer Possible Possible
Join tables Possible Not in real QBE (possible in SQL)

Auto. Mode possible with InnoDB
table using foreign key and new
« pmadb option »

Make some calcul (SUM,MIN,AVG…) Possible Not in real QBE (possible in SQL)
Update records with criterias Possible Not Possible (Possible with SQL)
Delete records with criteras Possible Not Possible (Possible with SQL)

Query with SQL

Both products let you write SQL query but Paradox  help you with a modern Visual Query Builder :

Note : If you try to use directly the SQL code created by Paradox  from a QBE (Menu View/View SQL) it
will not work since the alias is not  used.  You ‘ll have to select the SQL alias in the Alias listbox in the
SQL toolbar. Sometimes an SQL alias is included inside the SQL text and it will not work either !

On the other side PhpMyadmin will simply let you write your SQL query or use the search fonction (« QBE »)
but this mode will not help you with the jointure between tables…



International Paradox Convention Francfort , 30th and 31st October, 2004

Paradox and MySql Page 8 sur 17 Côme de Christen
www.clairinfo.fr

SQL passthrough or not ?

In some cases Paradox will run your query locally (with the BDE SQL parser) :

- Mixing local and remote tables (an heterogeneous query) , QBE or SQL will always run locally
- More than one remote alias , QBE or SQL will always run locally
- In others case you can control how your query will run in the property dialog

If you can it is preferable to run the query remotely for better performance and in order to benefit of all MySQL
functionalities. With the local mode you are limited by the BDE possibilities.

Import / Export

Paradox can easily export a local table (DB,DBF) resulting from a query on any SQL Server (Menu File/Export)
Paradox can also execute a remote SQL query :  SELECT … INTO OUTFILE… (cf MySql documentation)
For the importation we will have to work in 2 pass, import the data in a local table and then copy the data to
MySql (or we could write an OPAL script reading a file line by line and inserting them in MySql), so not so
simple.

PhpMyAdmin is really good in this area :

Others good utilities of phpMyAdmin

Data Dictionary and print view of table

Paradox Forms and reports directly bounded to Mysql

From this point phpMyAdmin will be useless as we will explore some unique feature of Paradox. His ability to
build quickly forms and reports even on remote Sql data.

Forms

A first try with the expert (so without any manual change) and … hop it works and quite quickly !

Reports

A first test with the expert without any change and one minute after…

But then can we mix our MySql tables with some Paradox one ? … Well … yes !
And use some calculated field ? … Well … yes !

Using SQL in OPAL on MySql tables

Minimal example

Var
   Db Database
   daParams DynArray[] String
   Q1 Sql
EndVar

daParams["username"] = "root"
daParams["password"] = ""

if not db.open("TestMysql",daParams) then
   MsgStop("Error !","Can’t open Database")
   Return
endif

Q1 = Sql

SELECT * FROM mytable

EndSql

if not Q1.executeSQL(db,":Priv:Reponse") then
   MsgStop("Error !","Can’t execute Query 1")
   return
endif

if not db.close() then
   MsgStop("Error !","Can’t close Database")
   Return
endif

endMethod

Others important points

We can centralize recurrent code in a library

LibSQL.lsl

Var
   db Database
endVar

method ConnectSQL() Logical
Var
   daParams Dynarray[] String
EndVar

daParams["username"]="root"
daParams["password"]=""

if not db.open("TestMysql",daParams) then
   errorshow("Erreur ! Impossible d'accéder à la base")
   return False
else
   return True
endif

method CloseSQL() Logical
if not db.close() then
   errorshow("Impossible de fermer la base")
   Return False
else
   Return true
endif
endMethod

method ExecSQL(sSQL String) Logical
Var
   Q1 SQL
EndVar

Q1 = Sql

~sSQL

EndSQL

if not executeSQL(db,Q1,":Priv:Reponse") then
   errorshow("Erreur ! Impossible d'exécuter la requête.")
   return False
else
   return True
endif

endMethod

method GetUiDataFromSQL(sSQL String,vUi UiObject) Logical
Var
   Q1 SQL
   TcRep,TcUi TCursor
EndVar

Q1 = Sql

~sSQL

EndSQL

if not executeSQL(db,Q1,tcRep) then
   errorshow("Erreur ! Impossible d'exécuter la requête.")
   return False
endif

TcUi.attach(vUi)
if not TcUi.empty() then
   errorshow("Erreur ! Impossible d'initialiser le traitement.")
   return False
endif

if not TcRep.add(TcUi,True,False) then
   errorshow("Erreur ! Impossible d'exécuter la requête.")
   return False
endif

return True

endMethod

…

Using SQL query we can also work on local Paradox table populated on demand

On this working example we have copied a mysql table (sal) to a local Paradox one SAL.DB, that was the easy
part with the Paradox Copy utility. Now we want to build a form which can extract data from MySql on our local
Paradox table but also which can update all local change back to MySql…

At the form level we have in this example 

Var
   Lib Library
   sSql String
endVar

Uses ObjectPal
   "LibSQL.lsl"
endUses

method MasterUpdate()

sSQL="UPDATE sal SET
Civil = '"+Civil+"',
Nom = '"+Nom+"',
Prenom = '"+Prenom+"'
WHERE NoAgent = '"+NoAgent+"'"

if not Lib.ExecSQL(sSQL) then
MsgStop("Erreur !","Impossible de mettre à jour les données sur le serveur")

   return
endif

endMethod

method MasterInsert()

sSQL="INSERT INTO sal (NoAgent,Civil,Nom,Prenom)
VALUES ('"+NoAgent+"','"+Civil+"','"+Nom+"','"+Prenom+"')"

if not Lib.ExecSQL(sSQL) then
   MsgStop("Erreur !","Impossible de mettre à jour les données sur le serveur")
   return
endif
endMethod

method MasterDelete()

sSQL="DELETE FROM sal WHERE NoAgent = '"+NoAgent+"'"

if not Lib.ExecSQL(sSQL) then
   MsgStop("Erreur !","Impossible de supprimer les données sur le serveur")
   return
endif

endMethod

method action(var eventInfo ActionEvent)

if eventInfo.isPreFilter() then
;// Ce code s'exécute pour chaque objet de la fiche

else
;// Ce code s'exécute seulement pour la fiche
   actionID = eventInfo.id()
   switch
      case actionID = dataUnlockRecord :
         switch
         case self.recordStatus( "Modified" ) and not self.recordStatus( "New" ) :

         MsgInfo("Vue :","Mise à jour...")
         MasterUpdate()
         case self.recordStatus( "New" )     :
         MsgInfo("Vue :","Insertion...")
         MasterInsert()
         endSwitch
    case actionID = dataDeleteRecord  :
      MsgInfo("Vue :","Suppression...")
      MasterDelete()
  EndSwitch
endIf

endMethod

method close(var eventInfo Event)

if eventInfo.isPreFilter() then
;// Ce code s'exécute pour chaque objet de la fiche

else

;// Ce code s'exécute seulement pour la fiche
   Lib.CloseSQL()
   Lib.close()
  endif

endMethod

method menuAction(var eventInfo MenuEvent)

if eventInfo.isPreFilter() then
;// Ce code s'exécute pour chaque objet de la fiche

else
;// Ce code s'exécute seulement pour la fiche

  ;// Forcer la validation si fermeture de fiche
 ;// Without this code you will lose your last change on the form close !

   if eventinfo.id()=MenuCanClose and isEdit() then
      action(DataEndedit)
   endif

endif

endMethod

method open(var eventInfo Event)

if eventInfo.isPreFilter() then
; Ce code s'exécute pour chaque objet de la fiche.
else
; Ce code s'exécute juste pour la fiche.
   Lib.open(":Local:LibSQL")
    if not Lib.ConnectSQL() then
        MsgStop("Erreur !","Impossible de se connecter à la base...")
        return
    endif
endif

endmethod
method keyPhysical(var eventInfo KeyEvent)

if eventInfo.isPreFilter() then
;// Ce code s'exécute pour chaque objet de la fiche

if eventinfo.vChar()="VK_ESCAPE" then
      disabledefault
      NomRech.MoveTo()
endif
else
;// Ce code s'exécute seulement pour la fiche
endIf

endMethod

On the search input field level we have

method changeValue(var eventInfo ValueEvent)

DoDefault

sSQL="SELECT NoAgent,Civil,Nom,Prenom FROM sal WHERE Nom LIKE '"+NomRech+"%'"

if not Lib.GetUiDataFromSQL(sSql,LSAL) then
   MsgStop("Erreur !","Impossible d'obtenir les données")
   return
endif

LSAl.Home() ;// LSAL is the tableframe

endMethod

On the t ableframe LSAL level we have

method open(var eventInfo Event)
   LSal.empty()
endMethod

On the naviguation bar we have

Nothing special , from left to right :

method mouseClick(var eventInfo MouseEvent)
   active.action(DataBegin)
endmethod
method mouseClick(var eventInfo MouseEvent)
   active.action(DataPriorRecord)
endmethod
method mouseClick(var eventInfo MouseEvent)
   active.action(DataNextRecord)
endmethod
method mouseClick(var eventInfo MouseEvent)
    active.action(DataEnd)
endmethod
method mouseClick(var eventInfo MouseEvent)
if isEdit() then
   action(DataEndEdit)
else
   action(DataBeginEdit)

endif
endmethod
method mouseClick(var eventInfo MouseEvent)
   active.action(DataSearch)
endmethod
method mouseClick(var eventInfo MouseEvent)
   active.action(DataSearchNext)
endmethod
method mouseClick(var eventInfo MouseEvent)
If not isEdit() then
   action(DataBeginEdit)
endif
active.action(DataInsertRecord)
NoAgent.MoveTo()
Endmethod

method mouseClick(var eventInfo MouseEvent)
;Confirmation

if MsgQuestion("Suppression de l'enregistrement :","On continue ?")<>"Yes" then
   return
endif

;Traitement

action(DataBeginEdit)
active.action(DataDeleteRecord)
action(DataEndEdit)

endmethod
method mouseClick(var eventInfo MouseEvent)
close()
endmethod

On the key field NoAgent we have  (We do not allow changing our key value)

method changeValue(var eventInfo ValueEvent)
if Self<>"" then
   Disabledefault
   MsgInfo("Message :","Vous ne devez pas modifier ce champ.")
endif
endMethod

If working with Linux be careful with case sensitivity of MySql object …

If your SQL is  SELECT * FROM SAL  but your MySql table name is sal you’ll get a MySql error…

Links

Upsizing to client/server : ftp://ftp.corel.com/pub/Paradox/PdoxWin/UnIndexed/CSWHITE.ZIP
MySql documentation : http://dev.mysql.com/doc/mysql/en/index.html
Easyphp :  http://www.easyphp.org
MyOdbc :  http://www.mysql.com/products/connector/odbc/
PhpMyAdmin :  http://www.phpmyadmin.net/home_page/


